# Basic Excel Tools and Functions

Guowei Zhao

Aug 20, 2024

## 1 Computing the present value

Present value is the value of future cash flows discounted to the present at a certain discount rate. If we are given a total of 5 years, a discount rate of 10%, and a cash flow of $100 per year, we can use Excel to calculate the present value of this cash flow sequence.

First, we calculate the present value of each year's cash flow using the principle of present value. The formula for the present value of a single cash flow is:

$$Present\ value = \sum_{t=1}^{N} \frac{CF_t}{(1+r)^t}$$

The present value for each period is displayed in the Present Value column. The present value of all cash flows is then calculated by summing the present values of each period. The result, which is 379.08, is displayed in the cell that sums cells `C5:C9`, as shown in Figure 1.

Another way to calculate the present value is to use Excel's built-in `NPV()` function. Note that the `NPV()` function in Excel and the financial or investment NPV are not exactly the same, because the cash flow sequence shown in Figure 1 does not include the initial investment at the beginning of the period (i.e., the cash flow for period 0). I will address the calculation of the real financial NPV later. For our current goal, we can use the `NPV()` function to calculate the present value of a sequence of cash flows with 5 periods and a cash flow of $100 in each period. The formula is:`=NPV(\$B\$2, B5:B9)`,The syntax for the `NPV` function is:`NPV(rate, value1, [value2], ...)`,

- `rate`: The discount rate, which is 10% in this case.
- `value1, value2, ...`: The cash flows for each period. In Excel, you can directly select the range of cells containing the cash flows.

| | A | B | C | D |
|---|---|---|---|---|
| 1 | **COMPUTING THE PRESENT VALUE** | | | |
| 2 | Discount rate | 10% | | |
| 3 | | | | |
| 4 | **Year** | **Cash flow** | **Present value** | |
| 5 | 1 | 100 | 90.9091 | <-- =B5/(1+$B$2)^A5 |
| 6 | 2 | 100 | 82.6446 | <-- =B6/(1+$B$2)^A6 |
| 7 | 3 | 100 | 75.1315 | <-- =B7/(1+$B$2)^A7 |
| 8 | 4 | 100 | 68.3013 | <-- =B8/(1+$B$2)^A8 |
| 9 | 5 | 100 | 62.0921 | <-- =B9/(1+$B$2)^A9 |
| 10 | | | | |
| 11 | Net present value | | | |
| 12 | Summing cells C5:C9 | 379.08 | <-- =SUM(C5:C9) | |
| 13 | Using Excel's NPV function | $379.08 | <-- =NPV(B2,B5:B9) | |
| 14 | Using Excel's PV function | -379.08 | <-- =PV(B2,5,B9,0,0) | |

Figure 1: Enter Caption

In most calculations, the rate parameter uses the value from cell B2. To prevent the reference from changing when the formula is copied to other cells, we lock the cell by adding dollar signs ($) before the row and column labels. For example, `\$B\$2` locks both the row and column.

Using the formula `=NPV(\$B\$2, B5:B9)`, the result is 379.08, which is displayed in the cell labeled "Excel's NPV function." This result is the same as the one obtained using the previous method.

The final method is to use the `PV()` function. The difference between the `PV()` function and the `NPV()` function is that the `PV()` function is suitable for calculating the present value of a sequence of cash flows where the cash flow values are fixed for each period. To use the `PV()` function, enter the following formula in the results cell:`=-PV(\$B\$2, 5, B9, 0, 0)`,The syntax for the `PV()` function is: `PV(rate, nper, pmt, [fv], [type])`,

- `rate`: The discount rate, which is 10% in this case.
- `nper`: The total number of payment periods, which is 5 years.
- `pmt`: The payment made each period, which is $100 per year.
- `[fv]`: The future value, or a cash balance you want to attain after the last payment is made. If omitted, it is assumed to be 0.
- `[type]`: The timing of the payment, 0 for the end of the period (default), and 1 for the beginning of the period. In most cases, we use type = 0, which represents an ordinary annuity (end-of-period payments).

| | A | B | C |
|---|---|---|---|
| 1 | **COMPUTING THE VALUE OF A FINITE ANNUITY** | | |
| 2 | Periodic payment, C | 1,000 | |
| 3 | Number of future periods paid, n | 5 | |
| 4 | Discount rate, r | 12% | |
| 5 | Present value of annuity | | |
| 6 | Using formula | | |
| 7 | Using Excel's **PV** function | 3,604.78 | <-- =-PV(B4,B3,B2,0) |
| 8 | | | |
| 9 | **Period** | **Annuity payment** | |
| 10 | 1 | 1,000.00 | |
| 11 | 2 | 1,000.00 | |
| 12 | 3 | 1,000.00 | |
| 13 | 4 | 1,000.00 | |
| 14 | 5 | 1,000.00 | |
| 15 | | | |
| 16 | Present value using Excel's **NPV** function | 3,604.78 | <-- =NPV(B4,B10:B14) |

Figure 2: Enter Caption

Using the formula `=-PV(\$B\$2, 5, B9, 0, 0)`, the calculation result is 379.08, which is displayed in the cell labeled "Using Excel's PV function." This result is the same as the one obtained using the previous two methods.

The PV function is used with a negative sign because it adheres to the convention of financial calculations, where the direction of cash flow is indicated by the sign. In this case, simply changing the sign does not alter the meaning of the result.

# 2 Computing the value of a finite annuity

Annuities are financial instruments that generate a fixed cash flow each year. As previously mentioned, when calculating a series of equal annual cash flows, the results from using the `PV()` and `NPV()` functions are the same. This is illustrated in Figure 2, which shows a 5-year sequence of annual cash flows of $1,000 each ,with a discount rate of 123,604.78 as the `=NPV(B4, B10:B14)` function.

## 2.1 Computing the value of a finite growing annuity

There is also a type of annuity known as a growth annuity, where the cash flows increase at a fixed rate each year. When calculating the present value of this type of cash flow series, it is important to use the `NPV()` function rather than the `PV()` function. This is illustrated in Figure 3. To calculate a growth annuity in Excel, be sure to use the dollar sign ($) in the formula to lock in the growth rate, as the 6% growth rate remains constant from year to year..

| | A | B | C |
|---|---|---|---|
| 1 | COMPUTING THE VALUE OF A FINITE GROWING ANNUITY | | |
| 2 | First payment, C | 1,000 | |
| 3 | Growth rate of payments, g | 6% | |
| 4 | Number of future periods paid, n | 5 | |
| 5 | Discount rate, r | 12% | |
| 6 | Present value of annuity | | |
| 7 | Using formula | | |
| 8 | | | |
| 9 | **Period** | **Annuity payment** | |
| 10 | 1 | 1,000.00 | |
| 11 | 2 | 1,060.00 | <-- =B10*(1+$B$3) |
| 12 | 3 | 1,123.60 | <-- =B11*(1+$B$3) |
| 13 | 4 | 1,191.02 | <-- =B12*(1+$B$3) |
| 14 | 5 | 1,262.48 | <-- =B13*(1+$B$3) |
| 15 | | | |
| 16 | Present value using Excel's **NPV** function | 4,010.91 | <-- =NPV(B5,B10:B14) |

Figure 3: Enter Caption

| | A | B | C |
|---|---|---|---|
| 1 | **INTERNAL RATE OF RETURN** | | |
| 2 | **Year** | **Cash flow** | |
| 3 | 0 | -800 | |
| 4 | 1 | 200 | |
| 5 | 2 | 250 | |
| 6 | 3 | 300 | |
| 7 | 4 | 350 | |
| 8 | 5 | 400 | |
| 9 | | | |
| 10 | Internal rate of return | 22.16% | <-- =IRR(B3:B8) |

Figure 4: Enter Caption

# 3 Internal rate of return

## 3.1 IRR and 'Goal Seek'

The Internal Rate of Return (IRR) is the discount rate that makes the NPV of an investment project equal to zero. In Excel, you can calculate the IRR of a cash flow series using the built-in `IRR(values, [guess])` function. Simply select all the cash flow cells to use in the calculation. This is illustrated in Figure 4.

Another method to calculate the IRR is to combine the principle of IRR calculation with Excel's built-in functions. The IRR is defined as the discount rate




Figure 5: Enter Caption

that makes the NPV of an investment project equal to zero. Therefore, we need to find the discount rate that results in an NPV of zero. Excel's Goal Seek function can accomplish this. By providing the formula and the desired NPV result, Goal Seek can determine the appropriate discount rate, or IRR.

First, calculate the NPV using a discount rate of 12%, which results in an NPV of 215.01. Next, go to the 'What-If Analysis' tab in the menu bar and select the "Goal Seek" function. Set the parameters as follows:

- **Set cell**: B12, which is the cell containing the NPV calculation.
- **To value**: 0, because we want the NPV to be zero.
- **By changing cell**: B2, which contains the discount rate.

After running Goal Seek, the discount rate in cell B2 will be adjusted to achieve an NPV of zero. This process will yield the IRR, which in this case is 22.16%, as shown in Figures 5 and 6.

## 3.2 Multiple rates of return

One drawback of the IRR is that when the cash flow sign changes more than once, multiple IRRs can be calculated. This can be confusing for decision-makers. Therefore, it is necessary to calculate all possible IRRs to provide a complete picture for the decision-making process.

In the previous IRR calculation, we introduced the syntax of the `IRR()` function, which is `IRR(values, [guess])`. We did not explain the `[guess]` parameter. This optional parameter allows you to provide an initial guess for the IRR. The `IRR()` function will use this guess to find the closest IRR and output the result. If



| | A | B | C | D | E | F |
|---|---|---|---|---|---|---|
| 1 | INTERNAL RATE OF RETURN | | | | | |
| 2 | Discount rate | 22.16% | | | | |
| 3 | | | | | | |
| 4 | **Year** | **Cash flow** | | | | |
| 5 | 0 | -800 | | | | |
| 6 | 1 | 200 | | | | |
| 7 | 2 | 250 | | | | |
| 8 | 3 | 300 | | | | |
| 9 | 4 | 350 | | | | |
| 10 | 5 | 400 | | | | |
| 11 | | | | | | |
| 12 | Net present value (NPV) | 0.00 | | | | |
| 13 | | | | | | |



Figure 6: Enter Caption



| | A | B | C | D | E | F |
|---|---|---|---|---|---|---|
| 1 | | | MULTIPLE INTERNAL RATE | | | |
| 2 | Discount rate | 6% | | | | |
| 3 | NPV | -3.99 | <-- =NPV(B2,B7:B11)+B6 | | | |
| 4 | | | | | | |
| 5 | **Year** | **Cash flow** | | | | |
| 6 | 0 | -145 | | | | |
| 7 | 1 | 100 | | | | |
| 8 | 2 | 100 | | | | |
| 9 | 3 | 100 | | | | |
| 10 | 4 | 100 | | | | |
| 11 | 5 | -275 | | | | |
| 12 | | | | | | |
| 13 | **Identifying the two IRRs** | | | | | |
| 14 | First IRR | 8.78% | <-- =IRR(B6:B11,10%) | | | |
| 15 | Second IRR | 26.65% | <-- =IRR(B6:B11,30%) | | | |

Figure 7: Enter Caption

| | A | B | C | D | E | F | G | H |
|---|---|---|---|---|---|---|---|---|
| 1 | | | MULTIPLE INTERNAL RATES OF RETURN | | | | | |
| 2 | Discount rate | 6% | | | | | | |
| 3 | NPV | -3.99 | <-- =NPV(B2,B7:B11)+B6 | | | | DATA TABLE | |
| 4 | | | | | | | Discount rate | NPV |
| 5 | Year | Cash flow | | | | | | =B3 |
| 6 | 0 | -145 | | | | | 0% | |
| 7 | 1 | 100 | | | | | 3% | |
| 8 | 2 | 100 | | | | | 6% | |
| 9 | 3 | 100 | | | | | 9% | |
| 10 | 4 | 100 | | | | | 12% | |
| 11 | 5 | -275 | | | | | 15% | |
| 12 | | | | | | | 18% | |
| 13 | | | | | | | 21% | |
| 14 | | | | | | | 24% | |
| 15 | | | | | | | 27% | |
| 16 | | | | | | | 30% | |
| 17 | | | | | | | 33% | |
| 18 | | | | | | | 36% | |
| 19 | | | | | | | 39% | |

Figure 8: Enter Caption

no guess is provided, Excel will use a default value 10%. This process is illustrated in Figure 7.

In Figure 7, a series of cash flows is given, and the NPV of these cash flows is calculated using a 6% discount rate. Observing the characteristics of this cash flow, it is noted that the sign of the cash flow changes twice: from period 0 to period 1 and from period 4 to period 5. This results in two different IRR values.

To find these IRRs, we use the `IRR()` function with different initial guess values. Specifically, we use `=IRR(B6:B11, 10%)` and `=IRR(B6:B11, 30%)`. These functions yield two different IRRs: 8.78% and 26.65%, respectively.

## 3.3 Data Table and sensitivity analysis

If we also want to understand how the NPV of the cash flow series changes with different discount rates, we can use Excel's Data Table function to perform a sensitivity analysis. This involves varying the discount rate to observe the corresponding changes in the NPV. By doing so, we can see how sensitive the NPV is to changes in the discount rate

First, we calculate the NPV at a discount rate of 6% and place the formula in cell H5. Then, we select the entire area that includes both the discount rates and the corresponding NPVs. This process is illustrated in Figure 8.

Next, click the 'Data Table' function in the 'What-If Analysis' tab. In the dialog box that appears, select the cell containing the discount rate as the 'Column Input Cell.' The Data Table will then automatically replace the discount rate with




Figure 9: Enter Caption



| | A | B | C | D | E | F | G | H |
|---|---|---|---|---|---|---|---|---|
| 1 | | | MULTIPLE INTERNAL RATES OF RETURN | | | | | |
| 2 | Discount rate | 6% | | | | | | |
| 3 | NPV | -3.99 | <-- =NPV(B2,B7:B11)+B6 | | | | DATA TABLE | |
| 4 | | | | | | | Discount rate | NPV |
| 5 | Year | Cash flow | | | | | | |
| 6 | 0 | -145 | | | | | 0% | -20.00 |
| 7 | 1 | 100 | | | | | 3% | -10.51 |
| 8 | 2 | 100 | | | | | 6% | -3.99 |
| 9 | 3 | 100 | | | | | 9% | 0.24 |
| 10 | 4 | 100 | | | | | 12% | 2.69 |
| 11 | 5 | -275 | | | | | 15% | 3.77 |
| 12 | | | | | | | 18% | 3.80 |
| 13 | | | | | | | 21% | 3.02 |
| 14 | | | | | | | 24% | 1.62 |
| 15 | | | | | | | 27% | -0.24 |
| 16 | | | | | | | 30% | -2.44 |
| 17 | | | | | | | 33% | -4.90 |
| 18 | | | | | | | 36% | -7.53 |
| 19 | | | | | | | 39% | -10.27 |

Figure 10: Enter Caption

the other values in the rate column, calculating the corresponding NPVs for each discount rate. This process is illustrated in Figures 9 and 10.

It can be observed that as the discount rate increases from 6% to 9%, the NPV changes from positive to negative. This indicates that there is an IRR between 6% and 9%. Similarly, the NPV changes from positive to negative between 24% and 27%, indicating that there is another IRR between 24% and 27%.

The data table may not be very easy to understand or visually appealing. To improve clarity and aesthetics, we can visualize the results by plotting a graph based on the data table, as shown in Figure 11.




Figure 11: Enter Caption

Clearly, in a graph where the horizontal axis represents the discount rate and the vertical axis represents the NPV, the presence of two different points where the curve intersects the horizontal axis indicates the existence of two different IRRs.

# 4 Example to illustrate IRR vs. MIRR

Another shortcoming of the IRR is the assumption that all cash flows, whether positive or negative, are reinvested at the same rate. This assumption often does not hold true in practice. The Modified Internal Rate of Return (MIRR) addresses this issue by introducing two different interest rates: the financing cost and the reinvestment rate.

- **Financing Cost**: This is the interest rate paid by the investor to obtain the funds, representing the opportunity cost of investing capital at the beginning of the project.

- **Reinvestment Rate**: This is the rate at which future cash flows are expected to be reinvested. It is typically based on the company's weighted average cost of capital (WACC) or a reasonable market rate.

Computationally, the Modified Internal Rate of Return (MIRR) is calculated by discounting all negative cash flows at the financing rate (or safe rate) and

| | A | B | C | D | E | F | G | H | I | J |
|---|---|---|---|---|---|---|---|---|---|---|
| 1 | | | | EXAMPLE TO ILLUSTRATE IRR VS MIRR | | | | | | |
| 2 | | | | | | | | | | |
| 3 | | | | | | | | | | |
| 4 | | Assumptions: | | | | | | | | |
| 5 | | Finance rate = safe rate | | | 5% | | | | | |
| 6 | | Reinvestment rate = cost of capital | | | 10% | | | | | |
| 7 | | | | | | | | finance rate/safe rate | einvestment rate | |
| 8 | | Year | Cash flow | | | Year | Cash flow | PV | FV | |
| 9 | | 0 | (10,000) | | | 0 | (10,000) | (10,000) | | (109,773) |
| 10 | | 1 | 60,000 | | | 1 | 60,000 | | 87,846.00 | - |
| 11 | | 2 | (110,000) | | | 2 | (110,000) | (99,773) | | - |
| 12 | | 3 | 60,000 | | | 3 | 60,000 | | 72,600.00 | - |
| 13 | | 4 | - | | | 4 | - | | | - |
| 14 | | 5 | - | | | 5 | - | | | 160,446 |
| 15 | | | | | | | | | | |
| 16 | | IRR 1 | 0.00% | | | MIRR | 7.89% | <--=MIRR(C9:C14,E5,E6) | MIRR | 7.89% |

Figure 12: Enter Caption

compounding all positive cash flows at the reinvestment rate (or cost of capital) to determine the terminal value. This process is illustrated in Figure 12.

In cells H9 and H11, the cash flows are negative, and in cells I10 and I12, the cash flows are positive. We discount the two negative cash flows to the beginning of the period using a financing cost of 5% and sum them to get -109,773, which is placed in period 0. The positive cash flows are compounded at a reinvestment rate of 10% to get a terminal value of 160,466, which is placed in period 5. This converts the original cash flow sequence into a new sequence where -109,773 is invested at the beginning and 160,466 is received at the end.

We then use the `IRR()` function to calculate the IRR of this new cash flow sequence, which is 7.89%, as shown in cell J16.

Similarly, we can use the `MIRR()` function to calculate the MIRR directly from the original cash flow. The syntax is `MIRR(values, finance_rate, reinvestment_rate)`, where:

- `values` represents the cash flows over the periods,
- `finance_rate` is the cost of financing,
- `reinvestment_rate` is the rate at which positive cash flows are reinvested.

# 5 Flat payment schudule

Once we know the present value of a sequence of cash flows, how do we calculate the fixed cash flows for each period? For example, if you borrow $10,000 at an interest rate of 7% and plan to pay it back in 6 years with equal principal and interest payments, how much do you need to pay back each year?

We can use the `PMT()` function to calculate this. The syntax for the `PMT()` function is: `PMT(rate, nper, pv, [fv], [type])`

- `rate`: The interest rate per period.

| | A | B | C | D | E | F |
|---|---|---|---|---|---|---|
| 1 | **FLAT PAYMENT SCHEDULES** | | | | | |
| 2 | Loan principal | 10,000 | | | | |
| 3 | Interest rate | 7% | | | | |
| 4 | Loan term | 6 | <-- Number of years over which loan is repaid | | | |
| 5 | Annual payment | 2,097.96 | <-- =PMT(B3,B4,-B2,0,0) | | | |
| 6 | | | | | | |
| 7 | | | | | | |
| 8 | | **Year** | **Principal at beginning of year** | **Payment at end of year** | **Interest** | **Return of principal** |
| 9 | | 1 | 10,000.00 | 2,097.96 | 700.00 | 1,397.96 |
| 10 | | 2 | 8,602.04 | 2,097.96 | 602.14 | 1,495.82 |
| 11 | | 3 | 7,106.23 | 2,097.96 | 497.44 | 1,600.52 |
| 12 | | 4 | 5,505.70 | 2,097.96 | 385.40 | 1,712.56 |
| 13 | | 5 | 3,793.15 | 2,097.96 | 265.52 | 1,832.44 |
| 14 | | 6 | 1,960.71 | 2,097.96 | 137.25 | 1,960.71 |
| 15 | | 7 | 0.00 | 2,097.96 | 0.00 | 2,097.96 |
| 16 | | | | | | |

Figure 13: Enter Caption

- `nper`: The total number of payment periods.
- `pv`: The present value, or the total amount that a series of future payments is worth now (in this case, the amount borrowed).
- `[fv]`: The future value, or the cash balance you want to attain after the last payment is made. If omitted, it is assumed to be 0.
- `[type]`: The timing of the payment, 0 for the end of the period (default), and 1 for the beginning of the period.

Using the formula `=PMT(B3, B4, -B2, 0, 0)`, we can calculate the annual repayment amount, which is \$2,097.96, as shown in Figure 13.

Since this is an equal principal and interest repayment method, the repayment amount is the same for each installment. If we want to determine how much of each installment is allocated to the principal and how much is allocated to the interest, we can calculate the detailed results and list them in a table, as shown in Figure 13.

# 6 Using XIRR to compute the annualized internal rate of return

The IRR function assumes that each cash flow occurs exactly at the beginning or end of the period, or that each period is spaced exactly one year apart. However,

| | A | B | C |
|---|---|---|---|
| 1 | **USING XIRR TO COMPUTE THE ANNUALIZED INTERNAL RATE OF RETURN** | | |
| 2 | **Date** | **Cash flow** | |
| 3 | 1-Jan-14 | -1,000 | |
| 4 | 3-Mar-14 | 150 | |
| 5 | 4-Jul-14 | 100 | |
| 6 | 12-Oct-14 | 50 | |
| 7 | 25-Dec-14 | 1,000 | |
| 8 | | | |
| 9 | IRR | 37.19% | <-- =XIRR(B3:B7,A3:A7) |

Figure 14: Enter Caption

in reality, this is rarely the case. For example, if last year's cash flow occurred in February, but this year's cash flow occurs in January, the time interval is less than one year. In other cases, the interval might be more than one year.

To more accurately calculate the IRR for cash flows that occur at irregular time intervals, we can use the `XIRR` function. The `XIRR` function takes into account the specific dates on which the cash flows occur, providing a more precise internal rate of return. The syntax for the `XIRR` function is: `XIRR(values, dates, [guess])`, where `values` fo the cash flow, `dates` for the specific time series, `guess` for the estimated target IRR value. This is shown in Figure 14.

Similarly, when cash flows occur at irregular points in time, the calculation of NPV can be performed using the `XNPV` function. The syntax for the `XNPV` function is: `XNPV(rate, values, dates)`, the meaning of its parameters is the same as those in the previous function. This is illustrated in Figure 15.

| | A | B | C |
|---|---|---|---|
| 1 | **USING XNPV TO COMPUTE THE NET PRESENT VALUE** | | |
| 2 | Annual discount rate | 12% | |
| 3 | | | |
| 4 | **Date** | **Cash flow** | |
| 5 | 1-Jan-14 | -1,000 | |
| 6 | 3-Mar-15 | 100 | |
| 7 | 4-Jul-15 | 195 | |
| 8 | 12-Oct-16 | 350 | |
| 9 | 25-Dec-17 | 800 | |
| 10 | | | |
| 11 | Net present value | 16.80 | <-- =XNPV(B2,B5:B9,A5:A9) |

Figure 15: Enter Caption